

# Kurze Einführung in die KASH3-Syntax

Vorlesung Kryptographie Wintersemester 2005/2006

Im folgendem soll eine kurze Einführung in die KASH3-Syntax gegeben werden. Dies geschieht anhand von Beispielen, an denen die Befehle und zu beachtende Dinge erläutert werden.

## 1 Erste Schritte in KASH3

KASH3 ist ein Interpreter um sich der KANT V4-C-Bibliotheken zu bedienen. KANT V4 ist ein auf algebraische Zahlentheorie spezialisiertes Computeralgebra-System, was an der TU Berlin von der KANT-Gruppe entwickelt wurde.

KASH3 startet man, indem in die Shell der Befehl `kash3` eingegeben wird. Es erscheint dann ein Banner und ein Eingabeprompt `kash`. Verlassen kann man KASH3 wieder mit `quit`; . Nach dem Aufruf von KASH3 gelangt man in einen Interpreter, d.h. es können Befehle eingegeben und sofort ausgeführt werden. KASH3-Programme werden mit `Read("Dateiname");` eingelesen. Mit dem Befehl `LogTo("Dateiname");` wird alles, was nach diesem Befehl ausgegeben wird, genauso in der Datei "Dateiname" gespeichert. Mit Eingabe von `LogTo();` wird dieser Vorgang beendet. Führt man speicherintensive Berechnungen in KASH3 durch, so kann man KASH3 mit `Kash -m n` aufrufen, wobei `n` die Anzahl der Bytes ist, die für KASH3 reserviert werden sollen.

Die Syntax von KASH3 ist der von der Programmiersprache Pascal teilweise recht ähnlich. Eine erste Besonderheit ist, daß KASH3 sowohl imperative als auch funktionale Elemente enthält. Wir werden dies später an geeigneten Beispielen sehen.

Folgende Wörter oder Befehle sind in KASH3 reserviert und können nicht als Variablen benutzt werden:

`and, do, elif, else, end, fi, for, function, if, in, local, mod, not, od, or, repeat, return, then, until, while, quit.`

Ansonsten kann jedes Wort als Variable benutzt werden, wie z.B.

```
text := "Hello world";  
Hello world
```

oder auch

```
a := 3;  
3
```

wobei `:=` der Zuweisungsoperator ist. Folgt auf einer Eingabe ein Semikolon, so versucht KASH3 den Befehl auszuführen. Ein wichtiges Merkmal von KASH3 ist, daß man kein Typmanagement benötigt. Um ein kleines Beispiel zu geben, wird erst der Begriff Liste in KASH3 erklärt. Eine Liste in KASH3 ist eine endliche angeordnetes Tupel von beliebigen Elementen. Ein einfaches Beispiel ist

```
L := [1..10000];  
[1..10000].
```

Hier ist L eine Liste, die aus den ersten 10000 ganzen Zahlen besteht. Man kann durch L[k] auf das k-te Element von L zugreifen, sofern die Liste L auch soviel Element enthält:

```
L[312];  
312.
```

Ein anderes Beispiel für eine Liste ist

```
K := [1, "HALLO", ZX, i-> i*2, , [] ];  
[ 1, "HALLO", Univariate Polynomial Ring in x over Integer Ring ,  
function ( i ) ... end, , [] ].
```

Hier haben wir unterschiedliche Objekte in der Liste: eine ganze Zahl, ein ASCII-String, ein Polynomring, eine Funktion, eine Leerstelle in der Liste L und eine leere Liste. Möchte man den Polynomring aus der Liste K lesen, so schreibt man

```
PolyRing := K[3];  
Univariate Polynomial Ring in x over Integer Ring
```

und KASH3 hat dann den Polynomring in der Liste L unter der Variable PolyRing gespeichert. Mit einer Definition einer Variablen ist also der Typ automatisch festgelegt. Listen sind jedoch keine Mengen: So erhält man durch Eingabe von

```
L := [3, 3, 3, 4, 4, 5, 6];  
[ 3, 3, 3, 4, 4, 5, 6]
```

eine Liste mit Elementen, die mehrfach auftauchen. KASH3 kennt aber auch Mengen im mathematischen Sinne. Der Befehl hierfür ist Set. Für obiges Beispiel ergibt sich dann

```
L := Set([3, 3, 3, 4, 4, 5, 6]);  
[ 3, 4, 5, 6].
```

Möchte man wissen, ob das Element 5 in der Menge L liegt, so gibt man

```
5 in L;
```

ein und erhält

```
true.
```

Als nächstes wollen wir eine Funktion in KASH3 definieren. Das folgende Beispiel soll eine Liste von Zahlen in eine Liste von Buchstaben eines vorgegebenen Alphabetes umwandeln. Die Buchstaben der Worte kommen aus einem Alphabet A, das wir als Liste definieren:

```
A := ['A','B','C','D','E','F','G','H','I','J','K','L','M',
      'N','O','P','Q','R','S','T','U','V','W','X','Y','Z'];
ABCDEFGHIJKLMNPOQRSTUVWXYZ
```

Um zu sehen wie man Funktionen definiert soll folgendes Beispiel dienen:

```
NumbersToText := function(L,alphabet)
  return(Apply(L, i-> alphabet[i+1]));
end;
```

Der Funktionsname ist `NumbersToText` und als Parameter erhält die Funktion eine Liste `L` von ganzen Zahlen und eine Alphabet `alphabet`. Als nächstes kommt eine sogenannte Listenfunktion. Der Befehl `List` verlangt eine Liste und eine Funktion, hier also `L` und `i -> alphabet[i+1]`. Die Funktion `i -> alphabet[i+1]` bildet eine Zahl `i` auf das `i+1`-te Element im Alphabet ab. Sinnvollerweise sollten also die Einträge der Liste `L` nur um eins verminderte Positionen von Elementen des Alphabets `alphabet` sein. Der Befehl `List` geht alle Elemente der Liste `L` durch, bildet sie bezüglich der vorgegebenen Abbildung ab und gibt eine neue Liste zurück. Diese besteht dann aus den Bildern von `i -> alphabet[i+1]`. Dabei wird die Reihenfolge eingehalten, d.h. in der zurückgegebenen Liste ist das `i`-te Element das Bild vom `i`-ten Element der Liste `L`. Schließlich wird die so erhaltene Liste mit `return` zurückgegeben. Das bedeutet, wenn man den Aufruf und die Zuordnung

```
text := NumbersToText([7,0,11,11,14], A);
```

macht, so ordnet `KASH3` der Variable `text` das Ergebnis des Aufrufes der Funktions `NumbersToText` zu, nämlich den Wert

```
text;
"HALLO"
```

`KASH3` unterscheidet zwischen globalen und lokalen Variablen. Das nächste Beispiel soll deutlich machen, was eine lokale und was eine globale Variable ist:

```
Vigenere := function(block, key,alphabet)
local S,B,Cipher;
  S := TextToNumbers(key,alphabet);
  B := TextToNumbers(block, alphabet);
  Cipher := NumbersToText(Apply([1..Length(S)],
    i-> (S[i]+B[i]) mod Length(alphabet)), alphabet);
  return(Cipher);
end;
```

Zunächst haben wir wieder einen Funktionsnamen `Vigenere`. Die Funktion verlangt die Parameter `Block`, `key` und `alphabet`. Die Namen der Parameter sind wieder so gewählt, daß sie selbst-erklärend sind mit den Begriffen, die in der Vorlesung eingeführt wurden. Als lokale Variablen werden nun `S`, `B` und `Cipher` gewählt. Dies macht man durch den Befehl `local` deutlich, dem eine Aufzählung lokaler Variablen folgt. Das bedeutet, daß diese Variablen ihre Gültigkeit nur innerhalb der Funktion `Vigenere` haben. Gibt man z.B. folgendes in die Shell ein:

```

S;
so gibt KASH3
Error, Variable: 'S' must have a value
zurück. Wurde aber vorher
S := 3;
eingegeben, so würde man dann
S;
3

```

erhalten. Global heißt eine Variable, wenn sie wie oben definiert wurde, also nicht innerhalb einer Funktion. Innerhalb der Funktion `Vigenere` besitzt die Variable `S` den innerhalb der Funktion zugewiesenen Wert, also hier `S := TextToNumbers(key, alphabet)`. Beim Verlassen der Funktion hat `S` wieder seinen alten Wert 3. Würde `S` nicht in der Aufzählung der lokalen Variablen auftauchen, so hat `S` auch innerhalb der Funktion den selben Wert wie außerhalb. Wie man sieht können auch innerhalb einer Funktion wieder Funktionen aufgerufen werden. Allerdings müssen diese vorher definiert werden.

## 2 Funktional versus imperativ

Als nächstes werden wir einige weitere Operatoren auf Listen kennenlernen. Gleichzeitig wird immer eine imperative Variante angegeben.

Der Operator `Filtered` verlangt als Parameter eine Liste `L` und eine boolsche Funktion, z.B. die Funktion `i -> IsPrime(i)`, wobei `IsPrime(n) = false` ist wenn `n` keine Primzahl war, ansonsten wird `true` zurückgegeben. `Filtered` gibt eine neue Liste zurück, die aus allen Elementen `k` der Liste `L` besteht, für die `IsPrime(k) = true` gilt.

```

Filtered([1..50], i-> IsPrime(i));
[ 2, 3, 5, 7, 11, 13, 17, 19, 23, 29, 31, 37, 41, 43, 47 ].

```

Wir erhalten also eine Liste der Primzahlen  $n \leq 50$ . Die imperative Variante wäre hier

```

for xx in [1.. 50] do
  if IsPrime(xx) then
    Print(xx, " ");
  fi;
od;
2 3 5 7 11 13 17 19 23 29 31 37 41 43 47

```

Möchte man nun wissen, ob in der Menge `[2342..2444]` sich irgendeine Primzahl befindet, so gibt man folgendes ein:

```

ForAny([2342..2444], i-> IsPrime(i));
true.

```

```

xx :=2341; b := false;
2341

```

```

FALSE
repeat
  xx := xx+1;
  b := IsPrime(xx);
  until(b or xx > 2443);
b;
true

```

Wir wissen also jetzt, daß sich in der Liste [2342..2444] mindestens eine Primzahl befindet. Die kleinste bekommen wir mit

```

First([2342..2444], i-> IsPrime(i));
2347

```

oder

```

repeat
  xx := xx+1;
  b := IsPrime(xx);
  until(b or xx > 2444);
xx;
2347

```

Als nächstes möchten wir wissen, ob in der Liste  $K := [2, 4, 10, 12, 18, 20, 22]$  nur gerade Zahlen vorkommen, also

```

L := [2, 4, 10, 12, 18, 20, 22];
[ 2, 4, 10, 12, 18, 20, 22 ]
ForAll([2, 4, 10, 12, 18, 20, 22], j-> j mod 2 =0);
true

```

```

while xx <= Length(L) do
  b := (L[xx] mod 2 =0);
  xx := xx+1;
od;
b;
true

```

Eine weitere wichtige Listenoperation ist `ListSplit(L, n)`, welche eine gegebene Liste  $L$  in Teillisten der Länge  $n$  aufteilt. Ist nun  $n$  kein Teiler der Listenlänge  $\text{Length}(L)$  und gilt  $\text{Length}(L) = qn + r$  mit  $|r| < n$  und  $q \in \mathbb{N}$ , so gibt KASH  $q$  Listen der Länge  $n$  und eine Liste der Länge  $r$  zurück:

```

A1 := "1234567890ABCDEFGHIJKLMNPOQRSTUVWXYZ) (, .=! ?ab
                                     cdefghijklmnopqrstuvwxyz ";;
ListSplit(A1, 15);
[ "1234567890ABCDE", "FGHIJKLMNQRST", "UVWXYZ) (, .=! ?ab",
  "cdefghijklmnopq", "rstuvwxyz " ]

```

Hierbei unterdrückt das Doppelsemikolon hinter der ersten Eingabe die Ausgabe von A1. Benötigt man das  $n$ -fache kartesische Produkt einer gegebenen Teilmenge  $A$ , so erhält man dies durch die Eingabe von `Cartesian(A, ..., A)`; wobei  $A$   $n$ -mal innerhalb der runden Klammern erscheint. Bei zu großem  $A$  kann es jedoch sein, daß KASH3 zuviel Speicher belegt und abgebrochen wird. Um dies zu vermeiden, benutzt man obigen Befehl `ListSplit`. Wendet man diesen auf  $A$  an und teilt dadurch  $A$  z.B. in Teillisten  $A_1, A_2, A_3$ , so brechnet man jeweils das kartesische Produkt von `Cartesian(Ai, Aj, Ak)`; ( $i, j, k \in \{1, 2, 3\}$ ).

### 3 Chinesischer Restsatz und endliche Körper

Der chinesische Restsatz wird in der Kryptographie-Vorlesung öfters benötigt, z.B. kann man ihn beim Secret Sharing anwenden für Polynome oder bei RSA für ganze Zahlen. Deshalb soll hier kurz erläutert werden, wie man in KASH3 damit arbeiten kann.

Seien z.B. Zahlen  $z_1 = 17$ ,  $z_2 = 25$  und  $z_3 = 133$  vorgegeben. Der Chinesische Restsatz besagt nun, daß für  $n = z_1 \cdot z_2 \cdot z_3$  die Tatsache

$$\mathbb{Z}/n\mathbb{Z} \cong \mathbb{Z}/z_1\mathbb{Z} \times \mathbb{Z}/z_2\mathbb{Z} \times \mathbb{Z}/z_3\mathbb{Z} \quad (1)$$

gilt. Wenn man von ein Element in dem Produktring auf der rechten Seite von (1) das zugehörige Element von  $\mathbb{Z}/n\mathbb{Z}$  berechnen möchte, kann man dies mit dem Hauptsatz der simultanen Kongruenzen tun. Dies bedeutet also, wenn man z.B. für ein Element  $(6, 24, 99)$  des Produktringes ein  $\alpha \in \mathbb{Z}/n\mathbb{Z}$  berechnen möchte, so tut man das in KASH3 mit

```
ChineseRemainderTheorem([6, 24, 99], [17, 25, 133]);
20049
```

d.h. also das gesuchte Element in  $\mathbb{Z}/56525\mathbb{Z}$  war  $\alpha = 20049$  mit  $\alpha \equiv 6 \pmod{17}$ ,  $\alpha \equiv 24 \pmod{25}$  und  $\alpha \equiv 99 \pmod{133}$ .

Im RSA-Verfahren ist ein sogenannter RSA-Modul  $n$  vorgegeben, wobei  $n = pq$  das Produkt zweier ungerader Primzahlen  $p$  und  $q$  ist. Ist z.B.  $p = 309811$  und  $q = 4000037$ , so berechnet man die Ordnung der multiplikativen Gruppe von  $\mathbb{Z}/n\mathbb{Z}$ . Diese ist  $(p-1)(q-1) = t = 1239251153160$ . Ist nun ein  $b \in \mathbb{N}$  mit  $\text{ggT}(b, t) = 1$ , so läßt sich ein  $a \in \mathbb{N}$  berechnen mit  $x^{ab} \equiv x \pmod{n} \forall x \in \mathbb{Z}/n\mathbb{Z}$ . Ist z.B.  $b = 32321$  so berechnet man  $a$  in KASH3 mit

```
Xgcd(32321, 1239251153160);
1, extended by:
ext1 := 11349226241,
ext2 := -296
```

und erhält damit Zahlen  $a, c \in \mathbb{Z}$  mit  $at + cb = 1$ . Hierbei ist  $a = 11349226241$  und  $c = -296$ , was man durch

```
11349226241*b -296*t;
1
```

verifiziert. Oftmals ist es erforderlich mit endlichen Körpern zu rechnen. Ein Endlicher Körper  $k$  wird in KASH3 z.B. durch

```
k := FiniteField(2, 8);
Finite field of size 2^8
```

erzeugt, wie man ihn z.B. bei AES benötigt. Die schon bekannten Körper  $\mathbb{Z}/p\mathbb{Z}$  für eine vorgegebene Primzahl  $p$  erzeugt man mit `FiniteField(p, 1)`. Um Elemente direkt angeben zu können, benutzt man

```
k := FiniteField(31847, 2);
Finite field of size 31847^2
a:=k.1;
k.1
```

Man kann dann alle  $31847^2 - 1$  von Null verschiedenen Elemente durch Potenzen von  $a$  erzeugen, d.h.  $k^\times = \langle a \rangle = \{a^k \mid k = 1, \dots, p^2 - 1\}$ .

Ist z.B.  $p = 31847$ , so sei  $k := \mathbb{Z}/p\mathbb{Z}$ . Nun definiert man sich ein Polynom in  $k[T]$  wie folgt:

```
k := FiniteField(p, 1);
Finite field of size 31847
kT := PolynomialAlgebra(k);
Univariate Polynomial Ring over GF(31847)
T:=kT.1;
kT.1
f := T^5 + 202*T^4 + 2*T^3 + 6475*T + 1;
kT.1^5 + 202*kT.1^4 + 2*kT.1^3 + 6475*kT.1 + 1
```

Ist  $g$  ein zweites Polynom

```
g := T^3 + 203*T^2 + 31585*T + 1;
kT.1^3 + 203*kT.1^2 + 31585*kT.1 + 1
```

so kann man Elemente  $\alpha, \beta \in k[T]$  berechnen mit  $\alpha f + \beta g = 1$ , nämlich durch

```
Xgcd(f, g);
1, extended by:
  ext1 := 27363*kT.1^2 + 8628*kT.1 + 4864,
  ext2 := 4484*kT.1^4 + 18735*kT.1^3 + 27737*kT.1^2 + 25276*kT.1 + 26984
```

wobei  $\alpha = 27363T^2 + 8628T + 4864$  und  $\beta = 4484T^4 + 18735T^3 + 27737T^2 + 25276T + 26984$  ist. Dies verifiziert man durch

```
(27363*T^2 + 8628*T + 4864)*f +
(4484*T^4 + 18735*T^3 + 27737*T^2 + 25276*T + 26984)*g;
1.
```